Formal Verification of ABCD Parameters Based Models for Transmission Lines

Elif Deniz[1], Adnan Rashid[1], Osman Hasan[2], and Sofiène Tahar[1]

[1]Department of Electrical and Computer Engineering
Concordia University
Montreal, Quebec, Canada
{e_deniz,rashid,tahar}@ece.concordia.ca
[2]School of Electrical Engineering and Computer Science
National University of Sciences and Technology
Islamabad, Pakistan
osman.hasan@seecs.nust.edu.pk

Abstract. Transmission lines are widely used in many safety-critical applications, such as embedded systems (printed circuit boards) and power grids. They are generally represented as two-port networks and ABCD parameters are used to develop mathematical models (system of equations) capturing the relationship between the sending and receiving quantities, i.e., voltages and currents. In this paper, we propose to use higher-order-logic theorem proving for formally verifying the ABCD parameters based models of transmission lines. In particular, we use the HOL Light theorem prover to formalize transmission lines as systems consisting of these two-port network models, such as short, medium and cascaded, and various constraints on components of networks. Next, we formalize the Kirchhoff's voltage and current laws capturing dynamics of voltages and currents in these two-port networks. Moreover, we use these laws for formally verifying the ABCD parameters based models of transmission lines. To demonstrate the practical effectiveness of our proposed approach, we formally verify the ABCD parameter based medium transmission line model of a Wireless Power Transfer (WPT) system, which is widely used in electric vehicles and embedded devices.

Keywords: Transmission Lines, ABCD Parameters, Higher-Order Logic, Theorem Proving, HOL Light, Wireless Power Transfer

1 Introduction

Transmission lines [12] are used to transmit electrical energy from source to destination in the form of electromagnetic waves and are considered as essential components of power transmission/distribution networks. Based on their length, they are categorized as short, medium or long transmission lines. Moreover, to represent some scenarios, like impedance matching and operational flexibility, a single classification may not be adequate, necessitating the use of cascaded transmission lines, where these lines are connected in series, such as short-short or short-medium and vice versa [12]. This categorization of transmission lines enables us to model them using their equivalent lumped circuit representations and thus greatly facilitates a precise modeling of currents and voltages. Furthermore, the applications of these transmission lines extend across various domains including power distribution, medical imaging, telecommunications and aerospace. For instance, transmission lines can be used in the design of power distribution systems in aircrafts to ensure the reliable supply of electrical energy to various avionic subsytems, contributing to the overall functionality, safety and performance of the aircraft [15].

Transmission lines are generally represented as two-port networks that are electrical circuits comprising of lumped elements, such as resistors, capacitors and inductors, with pairs of terminals on sending and receiving ends, enabling a connection to the external networks [4]. These networks are characterized by their ABCD parameters, also known as chain or transmission line parameters ensuring a systematic modelling of the transmission lines [12]. In particular, the ABCD parameters enable developing the mathematical models (system of equations) of these networks capturing a relationship between the sending and receiving quantities, i.e., voltages and currents, by incorporating their characteristics like impedance and admittance. Moreover, these mathematical models can also be derived from the application of the physical laws, such as the Kirchhoff's Current Law (KCL) and the Kirchhoff's Voltage Law (KVL) on the two-port networks corresponding to the transmission lines [14].

In this paper, we propose to use the HOL Light theorem prover for the formalization of the ABCD parameters based models of transmission lines and their analysis. One of the main motivations for choosing HOL Light is the availability of the required theories of linear algebra, vector and matrices. Moreover, its underlying logic kernel has been verified in the CakeML project [11], which raises the level of soundness of our work.

The main contributions of the paper are as follows:

- Deep embedded formalization of transmission lines and their two-port network models along with related ABCD matrices and associated circuit models.
- Formal verification of the KCL and KVL capturing the dynamics the circuit models of the transmission lines using the formalized libraries of multivariate calculus [6] in the HOL Light theorem prover.
- Formal verification of the ABCD parameters based models of transmission lines (short, medium and cascaded) based on the KCL and KVL implementations verified in the last step.
- Formal verification of the T-shaped circuit model [3] for Wireless Power Transfer (WPT) [10] that is widely used in electric vehicles and embedded devices.

To the best of our knowledge, no formalization of the ABCD parameters based modeling of transmission lines exists in the literature and thus, for safety-critical applications this type of analysis in HOL theorem proving is utmost important.

The source code of our formalization is available for download at [1] and can be used by other researchers and engineers for analysis and further enhancements.

The rest of the paper is organized as follows: In Section 2, we provide preliminaries introducing the notion of ABCD parameters of transmission lines, as well as briefly describing the HOL Light theorem prover. In Section 3, we present the formalization of the ABCD parameters based models of transmission lines. Section 4 provides the formal verification of the short, medium and cascaded transmission lines captured by the ABCD matrix. We provide the formal analysis of a WPT system in Section 5. Finally, Section 6 concludes the paper.

2 Preliminaries

In this section, we present the fundamental concepts of modeling transmission lines through ABCD parameters and matrices. We also briefly describe the HOL Light theorem prover and associated functions and symbols that are necessary for understanding the rest of the paper.

2.1 ABCD Parameters of a Transmission Line

A two-port network refers to a system with two input and two output terminals. Figure 1 illustrates a generic two-port network, which is to be characterized using ABCD parameters.



Fig. 1: Two-port Network

In the context of the two-port transmission line model, port 1 is characterized by an input current, denoted as I_1, and a corresponding input voltage, V_1. The resulting output voltage and current at port 2 are labeled as V_2 and I_2, respectively. It is essential to note that the chosen current directions designate I_1 as entering and I_2 as leaving the two-port network. Here, V_1 and I_1 are dependent variables whereas V_2 and I_2 are considered independent. Let A, B, C and D be constants that characterize the above network. These *ABCD parameters* relate the input variables V_1 and I_1 as functions of the output variables V_2 and I_2 as follows:

$$V_1 = AV_2 \;\; BI_2$$

$$I_1 = CV_2 \;\; DI_2$$

The above equations can be written in matrix form:

$$\begin{bmatrix} V_1 \\ I_1 \end{bmatrix} = \begin{bmatrix} A & B \\ C & D \end{bmatrix} \begin{bmatrix} V_2 \\ I_2 \end{bmatrix}$$

These parameters play an important role in the analysis and comprehension of the transmission of electrical signals through intricate networks. It is also utilized to evaluate the performance of input, output voltage, and current within the transmission network.

2.2 HOL Light Theorem Prover

HOL Light [5] is an interactive theorem proving environment for the development of mathematical proofs using higher-order logic. It is implemented in Objective CAML (OCaml) [13], which is a variant of the ML (Meta-Language) functional programming language. The significance of theorem proving systems lies in their ability to verify general properties of software and hardware systems. For example, they can help in verifying and establishing the accuracy of diverse properties of a digital circuit, ensuring that the circuit operates as intended and complies with the specified design criteria. Furthermore, HOL Light provides a variety of automated proof procedures and proof assistants to assist users in guiding and completing their proofs. In addition, users have the flexibility to craft and implement their own personalized automation methods. Table 1 summarizes some HOL Light functions/symbols and their meanings that are frequently used in our proposed formalization.

Table 1: HOL Light Functions and Symbols

HOL Light Functions/ Symbols	Standard Functions/ Symbols	Description
`&a`	$\mathbb{N} \to \mathbb{R}$	Type casting from Natural numbers to Reals
`&num`	$\{0, 1, 2..\}$	Positive Integers data type
`Cx a`	$\mathbb{R} \to \mathbb{C}$	Type casting from Reals to Complex
`real`	$\mathbb{R}$	Real data type
`complex`	$\mathbb{C}$	Complex data type
`A ** B`	$\mathbf{AB}$	Matrix-matrix multiplication
`A ** x`	$\mathbf{Ax}$	Matrix-vector multiplication

In the rest of the paper, we will use standard mathematical notations rather than pure HOL Light syntax notations to facilitate the understanding of the paper for a non-HOL Light user. Readers interested in viewing the original HOL Light formalization can see the source code for our formalization at [1].

3 Formalization of the ABCD Parameters Based Models

A transmission line is used to carry electrical energy from a generating source to various distribution units by transmitting voltage and current waves from one end to the other. Transmission lines can be represented by two-port networks (circuit models having lumped elements). They can be single transmission lines, such as, short and medium lines based on their lengths or cascaded depending on the number of lines connected in series. Single transmission line models can further be categorised based on the arrangement of various lumped components of the circuits. For example, a short circuit representation, where a resistor R is connected in series to a capacitor Ca is commonly known as a *series impedance model* for the short transmission line. Similarly, medium transmission lines can be represented by their Π or T models based on the connection of various circuit components [4]. Therefore, it is sufficient to model single transmission lines that can further be used in the cascaded transmission lines, connected in series.

In our formalization, we present these transmission lines models, based on their type, as an enumerated type definition in HOL Light as:

define_type "tl_models = ShortTL_SerImp | ShortTL_ShuAdm |
MediumTL_TCir | MediumTL_PiCir"

Next, in order to capture an interaction between various components and their impact on the equivalent circuit models, we need to model the voltage and current functions and their associated representations in the mathematical model. In this regard, we first model the types of the voltage and current functions, and the complex vectors and matrices using the type abbreviation based on the available types in HOL Light as follows:

new_type_abbrev ("vol_fun",':($\mathbb{C} \rightarrow \mathbb{C}$)')
new_type_abbrev ("cur_fun",':($\mathbb{C} \rightarrow \mathbb{C}$)')
new_type_abbrev ("comp_vec",':$\mathbb{C}^2$')
new_type_abbrev ("comp_mat",':${\mathbb{C}^2}^2$')

The type vol_fun is used to model a voltage function Vx that accepts a variable x of complex type $\mathbb{C}$ and returns a complex-valued function V. Similarly, the type comp_mat captures a 2×2 matrix having every entry as a complex number.

Similarly, we model the ABCD parameters and transmission line constants as 4-tuples in HOL Light. Moreover, we model the types of the sending and receiving end quantities as follows:

new_type_abbrev ("abcd_param",':(A $\times$ B $\times$ C $\times$ D)')
new_type_abbrev ("trans_lines_const",':(R $\times$ L $\times$ Ca $\times$ G)')
new_type_abbrev ("send_end_quan",':(Vs $\times$ Is)')
new_type_abbrev ("recei_end_quan",':(VR $\times$ IR)')

where the types of A, B, C, D, R, L, Ca, G, Vs, Is, VR and IR are given in Table 2. A transmission line is valid if it is represented by a valid ABCD parameters based model and satisfies various constraints on transmission line constants.
We formalize the validity of a transmission line in HOL Light as follows:

Definition 1. Valid Transmission Line
$\vdash_{def}$ $\forall$tlc tlm. valid_transmission_line (tlm,tlc) $\Leftrightarrow$
valid_transm_line_model tlm $\wedge$ valid_tl_const tlc

where the function valid_transm_line_model accepts a transmission line model, i.e., short or medium and provides its validity. This is formalized as:

Table 2: Parameters, and their Standard and HOL Light Symbols

Parameter Description	Standard Symbol	HOL Light Symbol:Type	Parameter Description	Standard Symbol	HOL Light Symbol:Type
Parameter A	A	A:$\mathbb{R}$	Parameter B	B	B:$\mathbb{R}$
Parameter C	C	C:$\mathbb{R}$	Parameter D	D	D:$\mathbb{R}$
Resistance	R	R:$\mathbb{R}$	Inductance	L	L:$\mathbb{R}$
Capacitance	Ca	Ca:$\mathbb{R}$	Conductance	G	G:$\mathbb{R}$
Sending End Voltage	V_s	Vs:vol_fun	Sending End Current	I_s	Is:cur_fun
Receiving End Voltage	V_R	VR:vol_fun	Receiving End Current	I_R	IR:cur_fun

Definition 2. Valid Transmission Line Models

$\vdash_{def}$ (valid_transm_line_model (ShortTL_SerImp) $\Leftrightarrow$ T) $\wedge$
(valid_transm_line_model (ShortTL_ShuAdm) $\Leftrightarrow$ T) $\wedge$
(valid_transm_line_model (MediumTL_TCir) $\Leftrightarrow$ T) $\wedge$
(valid_transm_line_model (MediumTL_PiCir) $\Leftrightarrow$ T)

Similarly, the function valid_tl_const in Definition 1 models the validity of the transmission line constants as follows:

Definition 3. Valid Transmission Line Constants

$\vdash_{def}$ $\forall$R L Ca G.
valid_tl_const ((R,L,Ca,G):trans_lines_const) $\Leftrightarrow$
&0 $<$ R $\wedge$ &0 $<$ L $\wedge$ &0 $<$ Ca $\wedge$ &0 $<$ G

The verification of a relationship between the sending and receiving end quantities (voltages and currents) ensures the correct working of the ABCD parameters based models of transmission lines. To verify this relationship, we first model the generalized ABCD matrix as follows:

Definition 4. Generalized ABCD Matrix

$\vdash_{def}$ $\forall$A B C D.

$$\text{abcd\_mat\_gen ((A,B,C,D):abcd\_param)} = \begin{bmatrix} \mathtt{A} & \mathtt{B} \\ \mathtt{C} & \mathtt{D} \end{bmatrix}$$

Next, we model the sending and receiving quantities as two-dimensional vectors in HOL Light:

Definition 5. Sending End Vector

$\vdash_{def}$ $\forall$Vs Is z.

$$\text{send\_end\_vec ((Vs,Is):send\_end\_quan) z} = \begin{bmatrix} \mathtt{Vs(z)} \\ \mathtt{Is(z)} \end{bmatrix}$$

Definition 6. Receiving End Vector
$\vdash_{def}$ ∀VR IR z.

$$\text{recei\_end\_vec}\ ((\text{Vs},\text{Is}):\text{recei\_end\_quan})\ \text{z} = \begin{bmatrix} \text{VR(z)} \\ \text{IR(z)} \end{bmatrix}$$

Now, we use Definitions 4, 5 and 6 to formalize a generalized relationship between the sending and receiving quantities of transmission lines as:

Definition 7. Relationship Between Sending/Receiving Quantities
$\vdash_{def}$ ∀Vs Is VR IR A B C D.
relat_send_receiv_quan_gen (Vs,Is) (VR,IR) (A,B,C,D) z ⇔
(send_end_vec (Vs,Is) z = ((abcd_mat_gen (A,B,C,D)):comp_mat) ∗∗
recei_end_vec (VR,IR) z)

Finally, we formalize KCL and KVL, capturing the dynamics of voltage and current in the circuit models, as follows:

Definition 8. KCL and KVL
$\vdash_{def}$ ∀cur_list z. kcl (cur_list:cur_fun list) (z:complex) =
((vsum (0..(LENGTH (cur_list) - 1)) (λn. EL n cur_list z)) = Cx (&0))
$\vdash_{def}$ ∀vol_list z. kvl (vol_list:vol_fun list) (z:complex) =
((vsum (0..(LENGTH (vol_list) - 1)) (λn. EL n vol_list z)) = Cx (&0))

where the functions kcl and kvl accept lists of currents and voltages across the components of circuits, cur_list and vol_list, a complex variable z, and return the implementations of KCL and KVL, respectively. For example, kcl ensures that the sum of all currents leaving a particular node is zero. Here, the function vsum accepts a vector-valued function f and provides the summation $\sum_{i=0}^{n} fi$. Similarly, the function `EL n l` extracts the n^{th} element from a list l.

4 Formalization of the Transmission Lines Models

This section presents the formal verification of the ABCD parameters based models of transmission lines, such as short, medium and cascaded transmission lines. To this end, we first formalize the ABCD matrices for the lumped circuits models in HOL Light as follows:

Definition 9. ABCD Matrices of Transmission Lines Models
$\vdash_{def}$ ∀R L Ca G w.
abcd_mat ShortTL_SerImp ((R,L,Ca,G):trans_lines_const) w =

$$\begin{bmatrix} 1 & \text{R ii} * \text{w} * \text{L} \\ 0 & 1 \end{bmatrix} \quad \wedge$$

abcd_mat ShortTL_ShuAdm ((R,L,Ca,G):trans_lines_const) w =

$$\begin{bmatrix} 1 & 0 \\ \frac{1}{\text{R}}\ \ \text{ii} * \text{w} * \text{Ca} & 1 \end{bmatrix} \quad \wedge$$

abcd_mat MediumTL_PiCir ((R,L,Ca,G):trans_lines_const) w =

$$\begin{array}{c} \begin{bmatrix} 1 & \dfrac{\mathsf{w * CaR + ii * w * L}}{2} & 0 \\ \dfrac{1}{\mathsf{R}} & \mathsf{ii * w * Ca} & 1 \end{bmatrix} \qquad \wedge \\ \color{blue}{\mathsf{abcd\_mat}}\ \mathsf{MediumTL\_TCir\ ((R,L,Ca,G){:}trans\_lines\_const)\ w} = \\ \begin{bmatrix} \mathsf{T}\left(1\ \dfrac{\mathsf{w * CaT}}{4}\right) & \mathsf{w * Ca} \\ \left(\dfrac{1}{\mathsf{R}}\right)\ +\ \mathsf{ii * w * Ca} & \left(1\ \dfrac{\mathsf{w * CaT}}{2}\right) \end{bmatrix} \end{array}$$

where $\mathsf{T = R + ii * w * L}$. The function abcd_mat accepts the model, transmission line constants and a variable w, and returns the ABCD matrix corresponding to the given model. Here, the symbol "ii" is used to represent the imaginary number.

4.1 Formalization of Short Transmission Lines

Short transmission lines are implemented by their two circuit representations, namely, series impedance and shunt admittance circuit models. The series impedance model is represented by the circuit given in Figure 2.



Fig. 2: Short Circuit Series Impedance Transmission Line

Here, the impedance effect is obtained from a series connection of the circuit's components resistor and inductor, respectively. We apply KVL on the series impendence circuit representation to obtain the following equation:

$$V_S = V_R \;\; Z I_R \tag{1}$$

where Z represents the impedance of the circuit. Similarly, KCL for the series impedance circuit (Figure 2) is mathematically expressed as follows:

$$I_S = I_R \tag{2}$$

Next, the KVL and KCL representations of the series impedance model, i.e., Equations (1) and (2) can be expressed using the ABCD parameters as follows:

$$V_S = AV_R \;\; BI_R \tag{3}$$

$$I_S = CV_R \;\; DI_R \tag{4}$$

where $A = 1$, $B = Z$, $C = 0$ and $D = 1$.

We start the verification of this model by formalizing the KVL implementation of the circuit in HOL Light as:

Definition 10. KVL for Short Circuit Series Impedance Implementation
⊢$_{def}$ ∀Vs Is VR IR w z R L Ca G.
  kvl_implem ShortTL_SerImp ((Vs,Is):send_end_quan)
    ((VR,IR):recei_end_quan) ((R,L,Ca,G):trans_lines_const) (w:real) z =
  (kvl [(λz. Vs z); (λz. −(VR z)); resis_volt R (λz. −(IR z));
        induct_volt L w (λz. −(IR z))] z)

where the functions resis_volt and induct_volt model the voltages across the resistor and inductor, respectively and their formalization is given as Entries 1 and 2 of Table 3. The function kvl_implem mainly accepts a transmission line model (here, ShortTL_SerImp), input and output quantities and transmission line constants, and return the KVL implementation corresponding to the given model[1].

Table 3: Voltage and Currents Across Various Components

Components' Names	Formal Models
Voltage across Resistance	⊢$_{def}$ ∀R Ir. resis_volt R (Ir:cur_fun) = (λz. Ir z ∗ R)
Voltage across Inductance	⊢$_{def}$ ∀L w Il. induct_volt L w (Il:cur_fun) = (λz. ii ∗ w ∗ L ∗ Il (z))
Series Impedance Voltage	⊢$_{def}$ ∀R L w Isi. series_imped_volt R L w (Isi:cur_fun) = (λz. (resis_volt R Isi) z + (induct_volt L w Isi) z)
Admittance Current	⊢$_{def}$ ∀Ca w Va. admit_curr_med Va Ca w = (λz. w ∗ $\frac{\text{Ca}}{2}$ ∗ (V z))

Our next step is to verify the equivalence of our formalization of the KVL for the short circuit series impedance implementation to its mathematical representation, which is captured as the following HOL Light theorem.

[1] Here, we only present the implementation of the short circuit series impendence model for brevity.

Theorem 1. KVL for Short Circuit Series Implementation Equivalence
$\vdash_{thm}$ ∀Vs Is VR IR w z R L Ca G.
let se = ((Vs,Is):send_end_quan) and
re = ((VR,IR):recei_end_quan) and
tlc = ((R,L,Ca,G):trans_lines_const) in
[A] valid_transmission_line (ShortTL_SerImp,tlc)
⇒ kvl_implem ShortTL_SerImp se re tlc w z ⇔
(Vs z = ((series_imped_volt R L w IR) z) + VR z)

where the function series_imped_volt (given as Entry 3 of Table 3) models the voltage across the series combination of the resistor and inductor, as depicted in Figure 2. The only assumption A provides the validity of the transmission line for the short circuit series impedance model. The HOL Light proof for the verification of Theorem 1 is given as follows[2]:

```
let SHORT_TL_SERIES_IMPED_KVL = prove(
‘∀Vs Is VR IR Ca G L R w z.
  kvl_implem ShortTL_SerImp ((Vs,Is):send_end_quan)
   ((VR,IR):recei_end_quan)((R,L,Ca,G):trans_lines_const) w z =
     (Vs z = ((series_imped_volt R L w (IR:cur_fun)) z) + VR z)‘,

REPEAT GEN_TAC THEN SIMP_TAC [series_imped_volt] THEN
  REWRITE_TAC [kvl_implem; kvl] THEN REWRITE_TAC [LENGTH] THEN
  SIMP_TAC [GSYM ONE; GSYM TWO; GSYM THREE] THEN
  SIMP_TAC [SUC_SUB1] THEN SIMP_TAC [VSUM_4_LIST] THEN
  SUBGOAL_THEN ‘induct_volt L w (λz. --(IR:cur_fun) z) z =
                 --(induct_volt L w (λz. IR z) z)‘ ASSUME_TAC
  THENL [REWRITE_TAC [induct_volt] THEN CONV_TAC COMPLEX_FIELD;ALL_TAC]
  THEN ASM_SIMP_TAC[] THEN POP_ASSUM (K ALL_TAC) THEN
  SUBGOAL_THEN ‘resis_volt R (λz. --(IR:cur_fun) z) z =
                 --(resis_volt R (λz. IR z) z)‘ ASSUME_TAC
  THENL [REWRITE_TAC [resis_volt] THEN CONV_TAC COMPLEX_FIELD;ALL_TAC]
  THEN ASM_SIMP_TAC[] THEN POP_ASSUM (K ALL_TAC) THEN
  SIMP_TAC [ETA_AX] THEN CONV_TAC COMPLEX_FIELD)
```

We start the proof by eliminating the universal quantifications using a repeated application of the tactic GEN_TAC. We use the tactical REPEAT to repeatedly apply a tactic (In this case, it is GEN_TAC). Subsequently, we simplfy our goal by employing our definition of series impedance voltage (Entry 3 of Table 3) through the tactic SIMP_TAC. We then utilize REWRITE_TAC to rewrite our goal (proof state) using functions about KVL (Definitions 8 and 10). Moreover, we continue the proof process by leveraging the theorems regarding number arithmetic, summation and lists, which are proven as a part of this formalization.

[2] Here, we only provide a detailed proof process of Theorem 1 and the details about verification of the rest of theorems can be found in our proof script [1].

It is important to note that the ASM variants of tactics REWRITE_TAC and SIMP_TAC provide rewriting and simplification tactics by incorporating the assumptions of the theorem to be proved. Finally, we use a decision procedure COMPLEX_FIELD, which automatically proves the basic field facts over complex numbers. Here, CONV_TAC is used to make a tactic from a conversion, which is COMPLEX_FIELD in our case. Additionally, we use the THEN-THENL structure to combine the whole proof, where the tactic THEN is used to combine two or more tactics applied to a goal. In case, if a tactic generates multiple subgoals *g1,..,gn*, one can utilize the THENL [*tac1;...;tacn*] to apply same set of tactics to different subgoals. Therefore, this structure provides a more compact representation of the proof as shown above in the HOL Light proof of Theorem 1.

Similarly, we formalize the KCL implementation of the short circuit series impedance implementation in HOL Light as follows:

Definition 11. KCL for Short Circuit Series Implementation
$\vdash_{def}$ ∀Vs Is VR IR w z R L Ca G.
 kcl_implem ShortTL_SerImp (Vs,Is) (VR,IR) (R,L,Ca,G) (w:real) z =
  (kcl [(λz. Is z); (λz. −(IR z))] z)

Next, we verify the equivalence of the above KCL implementation with its mathematical representation as the following HOL Light theorem.

Theorem 2. KCL for Short Circuit Series Implementation Equivalence
$\vdash_{thm}$ ∀Vs Is VR IR w z R L Ca G.
 let se = ((Vs,Is):send_end_quan) and
  re = ((VR,IR):recei_end_quan) and
  tlc = ((R,L,Ca,G):trans_lines_const) in
  [A] valid_transmission_line (ShortTL_SerImp,tlc)
   ⇒ kcl_implem ShortTL_SerImp se re w z tlc ⇔ (Is z = IR z)

The verification of the above theorem is very similar to that of Theorem 1.

Now, we verify a relationship between the sending and receiving quantities (voltages and currents) for the series impedance implementation of the short transmission line based on its KCL and KVL implementations as:

Theorem 3. Verification of the Short Transmission Line Model
$\vdash_{thm}$ ∀Vs Is VR IR w z R L Ca G.
 let se = ((Vs,Is):send_end_quan) and
  re = ((VR,IR):recei_end_quan) and
  tlc = ((R,L,Ca,G):trans_lines_const) in
  [A] valid_transmission_line (ShortTL_SerImp,tlc)
   ⇒ ((kvl_implem ShortTL_SerImp se re w z tlc) ∧
    (kcl_implem ShortTL_SerImp se re w z tlc)) ⇔
    (relat_send_receiv_quan ShortTL_SerImp se re tlc w z)

Assumption A ensures the validity of the short transmission line model. The proof process of Theorem 3 is mainly based on Theorems 1 and 2, and properties of matrices and vectors, alongwith some complex arithmetic reasoning.

We now formally verify the ABCD matrix based model of the series impendence circuit implementation of the short transmission line as:

Theorem 4. Verification of ABCD Matrix for Short Transmission Line
$\vdash_{thm}$ ∀Vs Is VR IR w z R L Ca G.
  let se = ((Vs,Is):send_end_quan) and
    re = ((VR,IR):recei_end_quan) and
    abcd = ((A,B,C,D):abcd_param) and
    tlc = ((R,L,Ca,G):trans_lines_const) in
    [A1] valid_transmission_line (ShortTL_SerImp,tlc) ∧
    [A2] A = 1 ∧ [A3] B = R + ii ∗ w ∗ L ∧ [A4] C = 0 ∧ [A5] D = 1
     ⇒ ((relat_send_receiv_quan_gen se re abcd z) =
       (relat_send_receiv_quan ShortTL_SerImp se re tlc w z))

Assumption A1 ensures that the series impedance model corresponds to a valid transmission line. Assumptions A2-A5 capture the ABCD parameters of the model. Finally, the conclusion of Theorem 4 ensures that the generalized ABCD parameters based transmission line model is equal to the short circuit series impedance implementation. The verification of the above theorem is mainly based on properties of vectors and matrices alongside some complex arithmetic reasoning. We also formally verify the shunt admittance circuit model for the short transmission lines and the details about its verification can be found in the HOL Light code available at [1].

4.2 Formalization of Medium Transmission Lines

The medium transmission lines are implemented by their two circuit representations, namely, the *Nominal* Π and T-circuit models, based on configurations of lumped components of the circuit. The nominal Π-circuit model for the medium transmission line is depicted in Figure 3.



Fig. 3: Π-circuit Medium Transmission Line

Here, the impedance effect is obtained from a series connection of the circuit's elements resistor and inductor, and the admittance effect is inherited from the capacitor. We start the verification of this model by formalizing the KVL implementation of the circuit in HOL Light as:

Definition 12. KVL for Medium Line Nominal Π Circuit Implementation
$\vdash_{def}$ ∀Vs Is VR IR.
kvl_implem MediumTL_NomPi (Vs,Is) (VR,IR) w z =
(kvl [(λz. Vs z); (λz. −(VR z)); resis_volt R (λz. −(IR z));
induct_volt L w (λz. −(IR z));
resis_volt R (admit_curr_med (λz. −(VR z)) Ca w);
induct_volt L w (admit_curr_med (λz. −(VR z)) Ca w)] z)

where the function admit_curr_med models the admittance current for the medium transmission line and its formalization is given as Entry 4 of Table 3.

Our next step is to verify the equivalence of the formalization of KVL for the nominal Π-circuit implementation to its mathematical representation, which is represented as the following HOL Light theorem.

Theorem 5. KVL for Medium Line Nominal Π Circuit Equivalence
$\vdash_{thm}$ ∀Vs Is VR IR R L Ca G w z.
let se = ((Vs,Is):send_end_quan) and
re = ((VR,IR):recei_end_quan) and
tlc = ((R,L,Ca,G):trans_lines_const) in
[A] valid_transmission_line (MediumTL_NomPi,tlc)
⇒ kvl_implem MediumTL_NomPi se re tlc w z ⇔

$$(\mathsf{Vs\ z} = \mathsf{VR\ z} + (\mathsf{R} + \mathsf{ii} * \mathsf{w} * \mathsf{L}) * \mathsf{IR\ z} + (\mathsf{R} + \mathsf{ii} * \mathsf{w} * \mathsf{L}) * \left(\frac{\mathsf{w} * \mathsf{Ca}}{2}\right) * \mathsf{VR\ z})$$

The only assumption A provides the validity of the transmission line for the nominal Π-circuit model. The proof process of Theorem 5 is mainly based on some properties of list and summation of complex-valued functions alongside some complex arithmetic reasoning.

Similarly, we formalize the KCL implementation of the nominal Π-circuit implementation. Here, we only show its implementation on the sending end only.

Definition 13. KCL for Medium Line Nominal Π-circuit Implementation
$\vdash_{def}$ ∀Vs Is Iz VR IR w z R L Ca G.
kcl_implem_s_end MediumTL_NomPi (Vs,Is)
(VR,IR) (Iz:cur_fun) w z (R,L,Ca,G) =
(kcl [(λz. Is z); (λz. −(Iz z)); admit_curr_med (λz. −(Vs z)) Ca w] z)

Next, we verify the equivalence of the above KCL implementation with its mathematical representation as the following HOL Light theorem.

Theorem 6. KCL for Medium Line Nominal Π-circuit Equivalence
$\vdash_{thm}$ ∀Vs Is VR IR Iz w z R L Ca G.
let se = ((Vs,Is):send_end_quan) and
re = ((VR,IR):recei_end_quan) and
tlc = ((R,L,Ca,G):trans_lines_const) in
kcl_implem_s_end MediumTL_NomPi se re (Iz:cur_fun) w z tlc ⇔
(Is z = Iz z + (admit_curr_med Vs Ca w) z)

The verification of the above theorem is very similar to that of Theorem 5. We further formalized the KCL on the receiving end of the model and combined both of these to obtain the final KCL implementation.

Next, we formally verify the ABCD parameters based representation of the nominal Π-circuit based model. We also formally verify the nominal T-circuit model for the medium transmission lines and the details about these formalizations are skipped here due to the space limitation and can be found in the HOL Light code available at [1].

4.3 Formalization of Cascaded Transmission Lines

When multiple transmission lines models are connected in series, it results into a cascaded transmission line as depicted in Figure 4. Here, n transmission lines are connected in series and a relationship between the sending and receiving end quantities is mathematically expressed as follows:



Fig. 4: Cascaded Transmission Line

$$\begin{bmatrix} V_S \\ I_S \end{bmatrix} = \begin{bmatrix} A_1 & B_1 \\ C_1 & D_1 \end{bmatrix} \cdots \begin{bmatrix} A_n & B_n \\ C_n & D_n \end{bmatrix} \begin{bmatrix} V_R \\ I_R \end{bmatrix} \tag{5}$$

where the ABCD matrix for the cascaded transmission line is a multiplication of matrices for each of the individual transmission line models. To formalize this ABCD matrix, we use the type abbreviation to model a collection of parameters for the individual transmission line models as:

```
new_type_abbrev ("w",`:ℝ`)
new_type_abbrev ("tl_models_param",`:(tl_models × trans_lines_const)`)
new_type_abbrev ("tl_models_param_all",`:(tl_models_param list × w)`)
```

where tl_models_param is a pair with its first element presenting a single transmission line model and the second element providing the transmission lines constants for the same transmission line present in the cascaded transmission line. Similarly, tl_models_param_all is a pair with its first element representing a list of parameters for all transmission line models present in the cascaded transmission line. The second element captures a variable w having a real data-type. Now, we use this type abbreviation to formalize the ABCD matrix for the cascaded transmission line in HOL Light as follows:

Definition 14. Cascaded ABCD Matrix
$\vdash_{def}$ ∀tlms tlmpa w R L Ca G.
  cascaded_abcd_matrix ([ ],w) = cidentity_mat ∧
  cascaded_abcd_matrix (CONS (tlms,(R,L,Ca,G)) tlmpa,w) =
    (abcd_mat tlms (R,L,Ca,G) w) ∗∗ cascaded_abcd_matrix (tlmpa,w)

where the function cascaded_abcd_matrix uses a variable of type tl_models_param_all and models the ABCD matrix of the cascaded transmission line using a recursive definition in HOL Light. We use the above definition to verify the ABCD matrix of a cascaded transmission line (medium and short transmission lines) as:

Theorem 7. ABCD Matrix of Cascaded Two Port Circuit/ Transmission Line
$\vdash_{thm}$ ∀R1 R2 L1 L2 Ca1 Ca2 G1 G2 w.
  let tlc1 = ((R1,L1,Ca1,G1):trans_lines_const) and
    tlc2 = ((R2,L2,Ca2,G2):trans_lines_const) in
    [A] valid_cascaded_tl ([MediumTL_PiCir,tlc1; ShortTL_SerImp,tlc2], w)
  ⇒ cascaded_abcd_matrix ([MediumTL_PiCir,tlc1; ShortTL_SerImp,tlc2], w) =

$$\begin{bmatrix} \mathsf{T3} & \mathsf{T3} * \mathsf{T2} + \mathsf{T1} \\ \mathsf{w} * \mathsf{Ca1} * \mathsf{T4} & \mathsf{w} * \mathsf{Ca1} * \mathsf{T4} * \mathsf{T2} + \mathsf{T3} \end{bmatrix}$$

where, $\mathsf{T1} = \mathsf{R1} \quad \mathsf{ii} * \mathsf{w} * \mathsf{L1}$, $\quad \mathsf{T2} = \mathsf{R2} \quad \mathsf{ii} * \mathsf{w} * \mathsf{L2}$

$$\mathsf{T3} = 1 \quad \frac{\mathsf{w} * \mathsf{Ca1} * \mathsf{T1}}{2} \qquad \mathsf{T4} = 1 \quad \frac{\mathsf{w} * \mathsf{Ca1} * \mathsf{T1}}{4}$$

The only assumption A ensures the validity of the cascaded transmission line. The proof process of the above theorem is mainly based on Definition 14, properties of lists and matrices, alongside some complex arithmetic reasoning. More details about its verification can be found in the code available at [1].

5 Application: Wireless Power Transfer System

A Wireless Power Transfer (WPT) system enables the transmission of electrical energy from source to destination without establishing a physical connection [7]. A WPT system uses the phenomenon of electromagnetic fields based on the induction coils to send energy from the transmitter to the receiver. It has been widely used in Electrical Vehicles (EVs) [2] and implantable medical devices [9]. For example, it is used for charging batteries in EVs, where the placement of wires is not possible due to a restricted space. To analyze the process of the energy transmission, the WPT system is represented as a T-shape transmission line lumped model [10]. Next, the ABCD parameters are analyzed to study the relationship between the voltages and current on the sending and receiving ends. The T-shape lumped medium transmission line model for the series-series compensation WPT system [10] is depicted in Figure 5. Here, R_1, C_1 and L_{k1} model the resistance, capacitance and leakage inductance on the sending end. Similarly, R_2, C_2 and L_{k2} capture the same quantities on the receiving end. M_{12} is the mutual inductance between the sending and receiving coils.

Similarly, V_1, I_1, V_2 and I_2 represent the sending and receiving end voltages and currents, respectively.



Fig. 5: T-Model for WPT System

Here, for simplicity, we consider the sending and receiving coils are identical. Therefore, the resistance, capacitance and leakage inductance on the sending and receiving ends are equal to each other, i.e., $R_1 = R_2 = R$, $C_1 = C_2 = C$ and $L_{k1} = L_{k2} = L$. Now, we can express a relationship between the sending and receiving quantities for T-shaped lumped model for the WPT system as follows [10]:

$$\begin{bmatrix} V_1 \\ I_1 \end{bmatrix} = \begin{bmatrix} 1 \ \frac{K_1}{K_2} & K_1 \left(2 \ \frac{K_1}{K_2} \right) \\ \frac{1}{K_2} & 1 \ \frac{K_1}{K_2} \end{bmatrix} \begin{bmatrix} V_2 \\ I_2 \end{bmatrix} \tag{6}$$

where,
$K_1 = R \ jwL_k \ \frac{1}{jwC}$, $K_2 = jwM_{12}$

We formally verify the above ABCD parameters based model of the WPT system (Equation (6)) based on the KCL and KVL implementations as the following theorem in HOL Light:

Theorem 8. ABCD Matrix of WPT System

```
⊢thm ∀Vs Is VR IR R Lk Ca G M12 w.
  let se = ((Vs,Is):send_end_quan) and
      re = ((VR,IR):recei_end_quan) and
      tlc = ((R,Lk,Ca,G):trans_lines_const) in
      [A] valid_transmission_line (MediumTL_TCir,tlc)
         ⇒ (kvl_implem_wpt MediumTL_TCir se re tlc w z ∧
            kcl_implem_wpt MediumTL_TCir se re tlc M12 w z) ⇔
                (relat_sr_wpt MediumTL_TCir se re tlc M12 w z)
```

The proof process of the above theorem is straightforward, thanks to our formalization of ABCD parameters based models presented in Sections 3 and 4. More details about its verification and the rest of the formalization can be found in the HOL Light code at [1].

Discussion

We introduced the first formalization of the ABCD parameters based models of transmission lines using an interactive theorem prover to ensure more sound and complete analysis. The main idea is to verify various aspects of transmission line designs, contributing to the development of formal models for power and communication systems. The utilization of HOL Light enabled us to verify generic properties, with universally quantified variables and thus can be reused and applied to different applications, such as printed circuit boards, coplanar waveguides, etc. Another distinguishing feature of our proposed formalization is the development of a systematic approach for analyzing the ABCD parameters based models of the transmission lines, which is easy to follow even for a non-HOL user. One of the challenging parts was the development of the deep embedding based formalization of the transmission lines, where we need to gather all details about the parameters that contribute to the dynamics of the two-port models of transmission lines and packaging them into our formalization. Moreover, it is worth mentioning that the use of higher-order-logic theorem proving in physical systems, particularly, transmission lines poses a significant challenge. This is mainly attributed to the considerable time and effort required for formalizing the underlying theories within higher-order logic. A potential approach to address this challenge involves the ongoing formal development of theories, encompassing libraries of practical and commonly used transmission lines in real-world applications. This approach has the potential to reduce the costs linked to incorporating theorem proving into the crucial stages of designing and verifying transmission line systems. The work presented in this paper can be considered as a foundational step, with the potential to verify numerous applications in related fields.

6 Conclusion

This paper proposed to use higher-order-logic theorem proving for formally analyzing the ABCD parameters based models of transmission lines. In particular, we formalized the lumped circuit models for the short, medium and cascaded transmission lines. Moreover, we formally verify the KVL and KCL implementations of these circuit models. Next, these implementations are used to verify the ABCD parameters based models of transmission lines. Finally, we used our proposed formalization for verifying the T-circuit model representation of a WPT system. In the future, we aim to establish a connection of these models with the telegrapher's equations [8] by formally analyzing the ABCD parameters based model for distributed (long) transmission lines. The formalization of long transmission lines requires the application of advanced mathematical techniques, such as matrix, complex variable analysis and differential equations.

References

1. Rashid, A and Deniz, E. https://hvg.ece.concordia.ca/code/hol-light/ABCD_Parameters/ABCD_Parameters.php (2024)
2. Bi, Z., Kan, T., Mi, C.C., Zhang, Y., Zhao, Z., Keoleian, G.A.: A Review of Wireless Power Transfer for Electric Vehicles: Prospects to Enhance Sustainable Mobility. Applied Energy 179, 413–425 (2016)
3. Davidson, C.W.: Transmission Lines for Communications. Springer (1978)
4. Glover, J.D., Sarma, M.S., Overbye, T.: Power System Analysis and Design, SI Version. Cengage Learning (2012)
5. Harrison, J.: Handbook of Practical Logic and Automated Reasoning. Cambridge University Press (2009)
6. Harrison, J.: The HOL Light Theory of Euclidean Space. Journal of Automated Reasoning 50, 173–190 (2013)
7. Imura, T., Imura, T.: Wireless Power Transfer. Springer (2020)
8. Inan, U.S., Inan, A.S., Said, R.K.: Engineering Electromagnetics and Waves. Pearson (2016)
9. Jiang, H., Zhang, J., Lan, D., Chao, K.K., Liou, S., Shahnasser, H., Fechter, R., Hirose, S., Harrison, M., Roy, S.: A low-frequency versatile wireless power transfer technology for biomedical implants. IEEE transactions on biomedical circuits and systems 7(4), 526–535 (2012)
10. Kim, D.: ABCD Parameters in Series-Series Compensation Wireless Power Transfer System. In: Sustainable Power and Energy Conference. pp. 1–5. IEEE (2022)
11. Kumar, R., Arthan, R., Myreen, M.O., Owens, S.: HOL with Definitions: Semantics, Soundness, and a Verified Implementation. In: Interactive Theorem Proving, LNCS, vol. 8558. pp. 308–324. Springer (2014)
12. Magnusson, P.C., Weisshaar, A., Tripathi, V.K., Alexander, G.C.: Transmission Lines and Wave Propagation. CRC Press (2017)
13. OCaml: https://ocaml.org/ (2023)
14. Saadat, H.: Power System Analysis. McGraw-Hill (1999)
15. Zelby, L.W., Mathes, J.B., Shawver, J.W.: Transmission Line Design for a Power Distribution System at 20 KHz for Aircraft. Tech. rep., NASA (1986)